Netherlands organization for applied scientific research

TNO-report

TNO Physics and Electronics Laboratory

P.O. Box 96864
2509 JG The Hague
Oude Waalsdorperweg 63
The Hague, The Netherlands

Fax +31 70 328 09 61
Phone +31 70 326 42 21

report no. FEL-91-B202

copy no. 9

title

An automatic meteorological station

author(s):

Ing. R. van der Touw

date :

November 1991

classification

title	: unclassified
abstract	: unclassified
report text	: unclassified
appendices A-B	: unclassified

no. of copies	: 15
no. of pages	: 34 (incl. appendices, excl. RDP & distr.list
appendices	: 2

TNO

report no.	:	FEL-91-B202
title	:	An automatic meteorological station
author(s)	:	Ing. R. van der Touw
institute	:	TNO Physics and Electronics Laboratory
date	:	November 1991
NDRO no.	:	-
no. in pow '91	:	
Research supervised by:		Dr.P.A.M. Jacobs
Research carried out by:		Ing.R. van der Touw

ABSTRACT (UNCLASSIFIED)

FEL-TNO owns an automatic weather station to support IR measurements in the field. After installation this station autonomously collects averaged data. The average is taken over the last 5 minutes interval and stored. The station can run unattended for at least a week. 25) * Meteorological instruments, Measuring instruments, Netherlands.

rapport no.	:	FEL-91-B202
titel	:	An automatic meteorological station
auteur(s)	:	Ing. R. van der Touw
instituut	:	Fysisch en Elektronisch Laboratorium TNO
datum	:	november 1991
hdo-opdr.no.	:	
no. in iwp '91	:	
Onderzoek uitgevoerd o.l.v.	:	Dr.P.A.M. Jacobs
Onderzoek uitgevoerd door	:	Ing.R. van der Touw

SAMENVATTING (ONGERUBRICEERD)

FEL-TNO beschikt over een automatisch weerstation om infrarood metingen te ondersteunen. Na installatie verzameld dit station autonoom weersgegevens. Dit zijn gemiddelden over het laatste 5 minuten interval. Het station kan minstens een week autonoom werken.

TABLE OF CONTENTS

1 INTRODUCTION

Results of measurements in the infrared wave band (IR) strongly depend on transmission of radiation through the atmosphere. In most cases an equipment like a transmissometer is not available at the measurement site. Not only the transmission depends on meteorological conditions, but also the thermal behaviour of materials. Solar heating for example has considerable influence on the surface temperature of sand, soil or roads. To study the temporal behaviour of targets and backgrounds in the IR wave band as a function of these parameters, it is necessary to register several meteo parameters during IR measurements. Therefore FEL-TNO combined high quality sensors, power supplies and a data acquisition system to establish an automatic meteorological station. The data, provided by the system, can be stored on disk and presented on screen, using a BASIC programme on an IBM-PC.

2 METEOROLOGICAL PARAMETERS

For future analysis it must be avoided to be dependent on other than standard sensors, to be able to make use of simple portable weather stations. Standard parameters are temperature, relative humidity, air pressure, wind speed and direction, hemispherical radiation (0.3 - 3 μm) and precipitation. An additional parameter that is valuable for IR research is hemispherical radiation in 3 - 50 μm. High quality sensors that measure these parameters are commercially available. Other parameters of interest to IR radiation measurements are cloudcoverage, cloudbase height, visibility and turbulence. The sensors to measure these parameters have limited possibilities and accuracies, if they are available anyway. They are very expensive and therefor seldom part of a standard weather observation set. The weather station described in this report contains all standard sensors plus a pyrgeometer to measure longwave hemispherical radiation (3 - 50 μm), in total eight different sensors. The sensor specifications are described in this chapter.

2.1 Air temperature and relative humidity

Air temperature and relative humidity are measured with a combined sensor, the Vaisala HMP 123Y. The temperature range is -25 °C to +45 °C, with a corresponding output of 0 to 20 mA. The sensor is a platinum resistance Pt-100, 100 Ω at 0 °C. The temperature accuracy is ±0.3 °C at 20 °C. The relative humidity is measured with a Humicap (capacitive humidity sensor) ranging from 0 - 100 % (0-20 mA output). The accuracy in the range 0 - 80 % is 2 %, from 80 - 100 % it is 3 %. The power input is 220 VAC, 50 Hz. The HMP 123Y is placed in a weather hut to be protected against direct radiation and precipitation (see photo 2.1).



Photo 2.1 Weather hut with HMP 123 Y and pressure sensor.

2.2 Air pressure

The air pressure is measured with a anaroid Thies baro-sensor, type 3.1150.10.000. It is specially designed for environmental conditions as -20 °C to +60 °C and high humidities. The sensor range is 946 to 1053 mbar with temperature compensation. The output are read over two resistance values (voltages). When α represents the fraction of the total pressure range ($0 \leq \alpha \leq 1$), the resistances are:

$R12 = 40 + 200\,\alpha - 160\,\alpha 2\ (\Omega)$

$R23 = 80 + 120\,\alpha - 160\,\alpha 2\ (\Omega)$

$R13 = 120.0\ (\Omega)$.

The sensor is placed in the weather hut (photo 2.1).

2.3 Wind speed and direction.

The wind speed sensor is from Thies, type 4.3303.22.000. This sensor gives 44 pulses per full rotation of the cups. Each puls is a 14 VDC signal. The sensor can be used to measure wind speeds ranging from 0.3 to 40 m/s and can be exposed to speeds up to 60 m/s without damage. The puls frequency at 40 m/s is 800 Hz. The puls frequency is linear proportional to the wind speed.

The wind direction sensor is also from Thies, type 4.3120.22.001. The wind direction is measured over 358°, with an accuracy of 0.5°, and transduced to resistance values. The 2° blind angle is at the South position. A full rotation covers resistance values starting from 90.0 Ω at 1° from South position (East side) to 155.0 Ω at North position and down to 40.0 Ω at 1° from South position at the West side.

Both wind sensors have internal heating elements. The operational temperature range is -35 to +60 °C. A Thies power supply unit, type 9.3386.00.000, is used for the wind speed sensor and heating elements. For power supply and data two separate cables are used. The maximum distance from sensors to power supply is 50 m. The sensors are mounted at 10 m level on a T-frame on top of a mast, including a lightning conductor. Photo 2.4 shows the sensors in the mast.



Photo 2.2 Wind mast with sensors and lightning conductor.

2.4 Radiation

2.4.1 Shortwave radiation

For this purpose a Kipp & Zonen pyranometer, type CM11 "solarimeter", is used. The CM11 is provided with a thermal detector. A thermopile is formed by about 200 thermocouples imprinted on a circular Al_2O_3 substrate (high thermal conductivity) using thick film techniques. The 100 hot junctions are near the centre in a rotational symmetric arrangment. The cold junctions are along the border, which is in good thermal contact with the case. When the pyranometer is illuminated there will be a heat flow from the centre to the border of the substrate in radial direction. This heat

flow is proportional to the absorptance of the black paint on the substrate (3M Velvet Black) and the irradiance.

Temperature dependency of some physical quantities, like conductivity, are compensated by a thermistor, applied in the electric circuit, to keep sensitivity constant. Doing so, the dependency of ambient temperature is limited to ± 1% over at least a temperature range of -10 °C to 40 °C.

The thermopile is protected against wind by two Schott K5 glass domes. The inner dome blocks the radiation exchange between thermopile and outer dome, necessary for a stable and small zero offset. The transmission of the glass limits the spectral range of the instrument to 0.308 - 2.730 μm. A dessicator in the housing prevents dew on the inner sides of the domes at cold nights.

A second not illuminated thermopile is installed to compensate for heat flows in the sensing element not caused by radiation.

The sensitivity at 20 °C is 4.86 $\mu V/Wm^{-2}$. Photo 2.3 shows the CM11 solarimeter (left on photo).

2.4.2 Longwave radiation

For measuring the longwave radiation an Eppley radiometer (pyrgeometer), type PIR (Precision Infrared Radiometer), is used. A silicon dome covers the thermopile and limits the spectral range to 4 - 50 μm. The output of the pyrgeometer is 4.56 $\mu V/Wm^{-2}$. The temperature compensation covers the range of -20 °C to +40 °C. In photo 2.3 the sensor is shown at the right side.



Photo 2.3 Radiation sensors on top of weather hut.

2.5 Precipitation

This sensor is from the manufacterer Lambrecht, type 1518H3. The principle is a balance with a volume of twice 2.0 ml, which tilts over after being filled by precipitation at one side. Each time the balance tilts over a reed-relay causes a pulse at the output. The measuring surface is 200 cm^2, so the resolution is 0.1 mm precipitation per square meter. The maximum intensity that can be measured is 7.5 mm/min.
The power supply, Lambrecht 1519H, is necessary for the heating to ensure proper functioning down to -25 °C. Without precipitation there is a constant 15 VDC signal on the output channel. Photo 2.4 shows the sensor.



Photo 2.4 Preciptation sensor.

2.6 Total weather station

The total weather station consists of four parts:

1. An airconditioned cabin in which the IBM-PC is placed, including the general power supply (220 VAC).
2. The mast with both wind sensors for wind direction and wind speed, both placed on a specially prepared top mast.
3. The main stand containing the weather hut with temperature, humidity, air pressure and radiation sensors and a box with power supplies and the Solartron S-net data aquisition system (photo 2.1)

The precipitation sensor is placed separately in the field

3 DATA ACQUISITION

The different sensors offer a variety of data formats. The data acquisition system that was chosen is a Schlumberger - Solartron Instruments "S-net". The system consists of two Isolated Measurement Pods (IMP), an analog (type 35951C) and a digital IMP (type 35952A), and a special Solartron Instruments adapter card in the PC, used to read the data from the IMPs and supply power to the IMPs. This is done by a single cable with a length of 100 m (maximum 1 km). Both IMPs can handle 20 channels. The analog IMP measurement ranges are DC voltages from 1 µV to 12 V (10 nA to 20 mA with 100 — shunt) and several thermocouple types. Results from all 20 channels are read in 1 second. The sample rates of the digital IMP are 20 Hz, 1, 10 and 100 kHz. The input functions are measurements of frequency (49 kHz max), periods (resolution 10 µs) and counts (24 bits). The digital IMP is used for the precipitation and the wind speed sensor, all the other sensors are connected to the analog IMP. The IMPs are closed black boxes (photo 2.1 below right).

3.1 Software programme meteo

To scan all sensors continuously a programme was written in BASIC language, called METEO. It runs within BASICA and uses Solartron Instruments subroutines to activate IMPs. Appendix A shows a listing of the programme. It starts measurements every 5 minutes, scanning all sensors during 4 minutes and 45 seconds. The remaining 15 seconds are used for calculations and to save the data. The programme saves and displays the averaged data of the last 5 minutes. The user can choose between display of averaged data or display the currently measured data by softkey input. In case of a power failure the programme restarts automatically. An output example is shown at Appendix B.

4 CABLES AND CONNECTORS

All the cables and connectors used in this weather station are described in this chapter. The corresponding codes can be found on the cables.

1. code: 220-MAINS
   Cable type : 3*1.0 mm^2.
   Connector : Souriau 840.25.132.001 and 840.25.870.501.
   Connections : 1 = 220 VAC (+), 2 = 220 VAC (-), 3 = ground.

2. code: 220-T+RH
   Cable type : 3*1.0 mm^2.
   Connector : Souriau 840.25.172.001 and 840.25.830.501.
   Connections : 1 = 220 VAC (+), 2 = 220 VAC (-), 3 = ground.

3. code: SD-PIR.RAD
   Cable type : 6*0.5 mm^2 shielded.
   Connector : Burndy MS-3120-E-12-10S and MS-3126-F-12-10P.
   Connections : A = low (pyrgeometer) = blue
   B = high (pyrgeometer) = red
   C = high (thermopile) = white
   D = thermistor = yellow
   E = thermistor = green
   F = ground = black
   Connected to S-net:
   IMP 35951C, Channel 13, high = red, low = blue (= ground).

4. code: MS-PRECIP

   Cable type : 4*1.0 mm².

   Connector : Burndy MS-3120-E-10-6S and MS-3126-F-10-6P.

   Connections : A = 42 VAC (+) = yellow/green

   B = 42 VAC (-) = brown

   C = ---

   D = contact = blue

   E = contact = black

   F = ---

   Connected to S-net:

   IMP 35952A, Channel 2, SP2 open, Input = green, Common = brown.

5. code: DATAREG, S-NET

   Cable type : 3 wired 16 AWG twisted pairs.

   Connector : Souriau 840.25.172.001 and 840.25.830.501.

   Connections : high = white, low = black, ground = shield.

   Connected to S-net:

   IMP 35952A : 120 Ω termination, grounded.

6. code: SP-WIND.DIR.

   Cable type : 3*0.75 mm² shielded

   Connector : Burndy MS-3130-E-12-3P and MS-3136-F-12-3S

   Connections : A = +10 VDC = green

   B = 0 VDC = bruin

   C = sliding-contact = white (yellow)

   Connected to S-net:

   IMP 35951C, Channel 9, high = green, low = yellow (ground = low).

   Channel 10, high = white, low = brown (ground = low).

7. code: SP-AIRPR.

   Cable type : 3*0.75 mm^2 shielded.

   Connector : Burndy MS-3130-E-12-3P and MS-3136-F-12-3S.

   Connections : 1 = +10 VDC = white (on connector:C)
   2 = sliding contact = green (on connector:A)
   3 = 0 VDC = brown (on connector:B)

   Connected to S-net:

   IMP 35951C, Channel 5, high = white, low = yellow (ground = low).
   Channel 6, high = green, low = brown (ground = low).

8. code: SD-T+RH.

   Cable type : 3*0.75 mm^2 shielded.

   Connector : Burndy MS-3130-E-12-3P and MS-3136-F-12-3S.

   Connections : 1 = relative humidity = green (A)
   2 = ground = brown (B)
   3 = temperature = white (C)
   4 = 220 VAC (+)
   5 = ---
   6 = 220 VAC (-)

   Connected to S-net:

   IMP 35951C, Channel 1, temperature,
   high = white, low = brown (ground = low)
   100 Ω shunt over high and low
   Channel 2, relative humidity,
   high = green, low = brown (ground = low)
   100 Ω shunt over high and low
   low (Channel 1) = low (Channel 2).

9. code: SD-CM11.RAD,

   Cable type : 4*0.5 mm^2 shielded.

   Connector : Burndy MS-3110-E-8-4S and MS-3116-F-8-4P.

   Connections : A = high = red (connector to IMP:green)

   B = ---

   C = low = blue (connector to IMP:brown)

   D = ground = white+black (connector to IMP:white)

   Connected to S-net:

   IMP 35951C, Channel 17, high = green, low = brown (ground = low).

10. code: MS-WIND.SP+Q,

    Cable type : 6*0.75 mm^2 shielded.

    Connector : Cannon CA-3101 and CA-3102 (20-7).

    Connections : A = +12 VDC = grey

    B = 0 VDC = yellow

    C = puls = green

    D = 26 VAC/3A (+) = pink

    E = 26 VAC/3A (-) = brown

    F = ---

    Connected to S-net:

    IMP 35952A, Channel 1, SP1 open, input = green, common = brown.

Figure 4.1 shows the connector pannel in the power supply box.

	O				
O	FUSE				
		O	O	O	O
220-MAINS	O	PIR	S-NET	AIRPR	CM11
O	0/1				
		O	O	O	O
220-T+RH		PRECIP	WIND.DIR	T+RH	WIND.SP+Q

Fig. 4.1 Lay-out connector pannel in power supply box

The codes on the cables consist of two parts. The first part indicates the signal flow, the second part describes the type of data. All codes are listed below and correspond with the used labels.

220-(type)	220 VAC to indicated sensor or power supply
MS-(type)	Mains power supply to Sensor
SP-(type)	Sensor to Power (to read resistance data)
PD-(type)	Power (data) to Data acquisition (S-net)
SD-(type)	Sensor to Data acquisition (S-net)
Datareg	Data registration on IBM-PC (S-net)
---	Ground
(type)=	
-MAINS	220 VAC to weather station
-+/-15V	Klaasing power supplies
-T+RH	Temperature and relative humidity
-AIRPR	Airpressure
-PRECIP	Precipitation
-WIND.DIR	Wind direction
-WIND.SP	Wind speed
-WIND.SP+Q	Wind speed and heating (both sensors)
-CM11.RAD	Shortwave radiation
-PIR.RAD	Longwave radiation

4.1 Connections at isolated measurement pods

The channels that are used of both IMPs are listed below.

IMP 35951C:

Channel	Parameter	High	Low/Ground	Remarks
01	Temperature	white	brown	100 Ω shunt
02	Rel Humid.	green	=	100 Ω shunt
05	Airpressure	white	yellow	
06	=	green	brown	green=yellow
09	Wind.dir	green	yellow	
10	=	white	brown	yellow=white
13	Pyrgeometer	red	blue	
17	Solarimeter	red	blue/black+white	to connector
	from connector	green	brown	to IMP

IMP 35952A:

Channel	Parameter	Output	Input	Common	Remarks
01	Wind.speed	-	green	brown	Rate 4
02	Precip	-	green	brown	

4.2 Additional power supply

Three connector blocks are mounted together in a box, as shown in fig. 4.2 One has 7 and two have 5 connections. The block with 7 connections has -15 VDC on the second, +15 VDC on the third and fourth (from left to right) and is used to supply the precipitation sensor. The blocks with 5 connections have both 10 VDC on the two most right connections and sensor output on the middle. They are used for air pressure and wind direction.



Fig. 4.2 Additional power supply box

5 CONCLUSIONS

To support IR measurements *FEL-TNO* combined high performance sensors to a dedicated automatic weather station. The software "METEO.BAS" performs the measurements, can handle errors and distinguishes incorrect values within certain ranges. It is sufficiently protected against moisture and dust and suitable for all European environmental conditions. The weather station can operate unattended for at least a week, depending on data storage capacity. During IR measurements a minimum of attention can be given to this station.

A.N. de Jong
(groupleader)

Ing. R. van der Touw
(author)

METEO Software listing

```
10 '
20 ' *************************************************************************
30 ' *****    *****    ****        ****          ****        ******     ******
40 ' *****  *  ***  *  ****  **************  ********  ***********  ***  *****
50 ' *****  **  *  **  ****  **************  ********  **********  *****  ****
60 ' *****  ***   ***  ****      **********  ********      ******  *****  ****
70 ' *****  *********  ****  **************  ********  **********  *****  ****
80 ' *****  *********  ****  **************  ********  ***********  ***  *****
90 ' *****  *********  ****        ********  ********        ******     ******
100 '**************************************************************************
110 '* Program for checking meteo-data with the Solartron S-net, display
120 '* these data on screen and write to disk (C\METEO\MMMDDYY.CAL),
121 '* using Solartron-Schlumberger subroutines.
130 '* These subroutines are:
140 '*   IMPINIT        INITialisation of the attached IMP's
150 '*   IMPTX          Transmits a Character string to an IMP
160 '*   IMPTEST        TESTs for data availability from IMP and stream
170 '*   IMPNUMERIC     Retrieves NUMERIC data from an IMP
180 '*   IMPSTRING      Retrieves character (STRING) data from an IMP
190 '*                  (Also numeric without changing the result format)
200 '*
210 '* DAD   = Driver ADdress
220 '* CAD   = Card ADdress       for IBM PC/XT : A000
230 '* IMPAD = IMP ADdress        analoog (1C) on 01 , digital (2A) on 02
240 '* ATT   = ATTached           IMP to adaptor card
250 '* FOUND = IMP FOUND          error indication : IMP attached or not
260 '* IMPER = IMP ERror
270 '* ADR   = ADdRess            IMP
280 '* STR   = STReam             type of data from IMP
290 '* IMPTX$= Character string (X) to be Transmitted to IMP
300 '* TXER  = Transmit Charater string (X) ERror
310 '* RXER  = State of IMP ERror
320 '* DAV   = Data AVailable     On IMP, 0 or 1
330 '* POD   = IMP number         1 or 2
340 '* CHA   = CHAnnel number     1 to 20
350 '* ACT   = ACTivated key F2   0 or 1
360 '* MT    = Meteo Tabel active            0 or 1
370 '* MK    = Meteo Key active              0 or 1
380 '* SCR   = SCReen full indication        0 or 1
390 '* KPR   = Key PRessed
400 '*
410 '* Timing Parameters
420 '* TIM      = FUNCTION converts time to seconds
430 '* DAY$     = FUNCTION converts date to string-date
440 '* DMY$     = Date in String format
450 '* MONTH$   = Name of the months for DMY$
460 '* START$   = TIME of next scan, first coming rounded 5 minutes
470 '* INTERVAL = Time interval in which scanning takes place
480 '* END.TIME = Sum of last two mentioned parameters
490 '* A,B      = Clock during scanning, before and after CALL
500 '* DELTA    = Difference between B and A : time per scan
510 '*
520 '* METEO parameters
530 '* SCAN        = SCAN number during INTERVAL
540 '* RES         = RESults            data per channel
550 '* RAW         = RAW data per channel, same as RES
560 '* STA         = STAtus             significance of data
570 '* SIGN        = SIGNificance of data per channel, same as STA
580 '* IMP.DATA    = (Sum of) RESults, data
590 '* STATUS      = Significance of data
600 '**************************************************************************
610 '**************************************************************************
620 '* COUNT       = Amount of additions per channel
630 '* NUMBER      = COUNT, fitting to METEO.ARRAY
```

```
640 '* CHANNEL$    = Text of measured quantities for RAW data display
650 '* METEO.ARRAY = METEO data, 8 different types
660 '*
670 '* --.CAL   = CALibration factor
680 '* --.CUR   = CURrent measured by IMP
690 '* --.VOL   = VOLtage measured by IMP
700 '* --.FRE   = FREquency
710 '* TEMP.-   = TEMPerature, .DEG for unity: DEGrees celcius
720 '* RHUM.-   = Relative HUMidity, .PRO for unity: percents
730 '* AIRP1.-  = AIRPressure of first channel, 5
740 '* AIRP2.-  = AIRPressure of second channel, 6
750 '* PRES.-   = airPRESsure in mbar, calculated per channel on .5 and .6
760 '* AIRP.BAR = Average AIRPressure in mbar
770 '* ANGLE1   = Wind direction of first channel, channel 9
780 '* ANGLE2   = Wind direction of second channel, channel 10
790 '* A1.-     = - of ANGLE1
800 '* A2.-     = - of ANGLE2
810 '* A.-      = Sum of all A1.- and A2.- of the same type (SIN or COS)
820 '* -.SIN    = SIN of ANGLE1 or ANGLE2
830 '* -.COS    = COS of ANGLE1 or ANGLE2
840 '* FWS      = Frequency of Wind Speed per scan
850 '* WS       = Wind Speed per scan for wind direction calculations
860 '* SX , SY  = Sum of all X (Y) components of WS*Ai.COS (WS*Ai.SIN)
870 '* WIND.DIR = Calculated WIND DIRection
880 '* PIR.-    = Precision Infrared Radiometer, .RAD for RADiance in W/m2
890 '* CM11.-   = Solarimeter CM11, .RAD for RADiance in W/m2
900 '* WISP.-   = WInd SPeed, .MPS for unity: Meters Per Second
910 '* FREQ     = FREQuency
920 '* RAIN.-   = RAIN, .MIN for unity: per MINute
930 '* PRECIP.MM= Amount of PRECIPitation during INTERVAL in MM
940 '* PULS     = Amount of PULSes measured by IMP
941 '* FILEM$   = METEO output file
942 '* FILET$   = THERMOCOUPLE output file
943 '* DRIVE$   = Current DRIVE for writing data
950 '***********************************************************************
960 CLS
970 KEY OFF
980 '
990 'LOGO DISPLAY
1000 '
1010 PRINT"    --------------------------------------------------------------
1020 PRINT
1030 PRINT"     MMMM      MMMM     EEEEEEEE     TTTTTTTTTT     EEEEEEEE      OOOOO
1040 PRINT"     MM MM    MM MM     EE              TT         EE           OO    O
1050 PRINT"     MM  MM MM  MM      EE              TT         EE          OO
1060 PRINT"     MM   MMM   MM      EEEEEE          TT         EEEEEE      OO
1070 PRINT"     MM         MM      EE              TT         EE          OO
1080 PRINT"     MM         MM      EE              TT         EE           OO    O
1090 PRINT"     MM         MM      EEEEEEEE        TT         EEEEEEEE      OOOOO
1100 PRINT
1110 PRINT"    --------------------------------------------------------------
1120 PRINT
1130 PRINT TAB(12);"Program Name";TAB(37);":     METEO.BAS"
1140 PRINT TAB(12);"Developed and written by :      Ing. R. van der Touw"
1150 PRINT TAB(12);"Date written";TAB(37);":     13 january 1990"
1160 PRINT TAB(12);"Last revised";TAB(37);":     13 january 1990"
1170 PRINT
1180 PRINT TAB(21);"Physics and Electronics Laboratory TNO"
1190 PRINT TAB(28);"Oude Waalsdorperweg 63"
1200 PRINT TAB(32);"P.O. Box 96864"
1210 PRINT TAB(30);"2509 JG The Hague"
1220 PRINT TAB(31);"The Netherlands"
1230 PRINT TAB(29);"Tel. (31) 70 3264221"
1240 '
```

```
1250 'DEFINE BLOCK
1260 '
1270 CLEAR , &HF000 , 1024
1280 DAD% = &HF000
1290 CAD% = &HA000
1300 ON ERROR GOTO 6140
1310 '
1320 DIM ATT%(2),IMPAD%(2),RES(20),STA%(20),MONTH$(12)
1330 DIM RAW(17),SIGN%(17),CHANNEL$(17)
1340 DIM IMP.DATA(20,2),STATUS%(20,2),COUNT%(20,2)
1350 DIM METEO.ARRAY(17),NUMBER%(17)
1360 '
1370 DEF FN TIM(TIM$)=3600*VAL(TIM$)+60*VAL(MID$(TIM$,4,2))+VAL(RIGHT$(TIM$,2)
1380 DEF FN DAY$(DAT$)=MID$(DAT$,4,3)+MONTH$(VAL(LEFT$(DAT$,2)))+RIGHT$(DAT$,5
1390 FOR A.INDEX=1 TO 12
1400     READ MONTH$(A.INDEX)
1410 NEXT
1420 DATA JAN,FEB,MAR,APR,MAY,JUN,JUL,AUG,SEP,OCT,NOV,DEC
1430 FOR J%=1 TO 17
1440    READ CHANNEL$(J%)
1450 NEXT
1460 DATA Temp(mA),RelHum(mA),Airpr(V),Airpr(V),Winddir(V),Winddir(V)
1470 DATA FarIR(V),NearIR(V),Windsp(c),Precip(c)
1471 DATA Temp 1.5(C),Temp -.1(C),Temp -.5(C),Temp conc(C)
1472 DATA Temp 5(C),Temp 7.5(C),Temp (PIR)
1480 ACT%=0
1485 FIN%=0
1490 INTERVAL=280
1500 GOTO 1520
1510 FOR I=1 TO 4000:NEXT
1520 FOR I=1 TO 1000:NEXT
1530 DMY$=FN DAY$(DATE$)
1531 DRIVE$ = "C:\METEO\DATA\"
1532 FILEM$=DRIVE$+MID$(DMY$,4,3)+LEFT$(DMY$,2)+RIGHT$(DMY$,2)+"M"
1533 OPEN FILEM$ FOR APPEND AS #1
1540 FILET$=DRIVE$+MID$(DMY$,4,3)+LEFT$(DMY$,2)+RIGHT$(DMY$,2)+"T"
1550 OPEN FILET$ FOR APPEND AS #3
1560 BLOAD"IMPDRIVE.MAC",DAD%
1570 IMPTX       = DAD%
1580 IMPTEST     = DAD%+3
1590 IMPNUMERIC  = DAD%+6
1600 IMPSTRING   = DAD%+9
1610 IMPINIT     = DAD%+12
1620 IMPAD%(1)=1
1630 IMPAD%(2)=2
1640 ATT%(1)=1
1650 ATT%(2)=1
1660 '
1670 ' IMP INITIALISATION
1680 '
1690 FOUND%=0
1700 CALL IMPINIT (CAD%,ATT%(1),FOUND%)
1710 IF FOUND%<>0  THEN IMPER%=1:LINEREP=1700:GOTO 6290
1720 '
1730 ' KEY FUNCTION BLOCK 1
1740 '
1750 ON KEY (1) GOSUB 1780
1760 KEY (1) ON
1770 GOTO 1810
1780 ON KEY (10) GOSUB 5940
1790 KEY (10) ON
1800 RETURN
1810 ON KEY (2) GOSUB 6000
1820 KEY (2) ON
```

```
1830 ON KEY (3) GOSUB 6060
1840 KEY (3) ON
1850 '
1860 ' IMP COMMANDS
1870 '
1880 A$="RE;AR;"
1890 B$="CH1MO500;CH2MO500;CH3MO360;CH4MO360;CH5MO100;CH6MO100;CH7MO360;CH8MO3
1900 C$="CH9MO100;CH10MO100;CH11MO360;CH13MO100;CH15MO360;CH16MO360;CH17MO100;
1910 D$="CH1MO740;CH1RA4;CH2MO740;"
1920 E$="CO"
1930 IMPTX$=A$+B$+C$+E$
1940 TXER%=0
1950 CALL IMPTX (IMPAD%(1),IMPTX$,TXER%)
1960 IF TXER%<>0 THEN IMPER%=2:LINEREP=1950:GOTO 6290
1970 IMPTX$=A$+D$+E$
1980 CALL IMPTX (IMPAD%(2),IMPTX$,TXER%)
1990 IF TXER%<>0 THEN IMPER%=2:LINEREP=1980:GOTO 6290
2000 KPR%=0
2010 '
2020 ' SCAN BLOCK
2030 '
2040 WHILE KPR%<>2 AND IMPER%=0
2050    IF KPR%=2 THEN 6640
2060    ADR%=0
2070    IMPTX$="TR"
2080    TXER%=0
2090    CALL IMPTX (ADR%,IMPTX$,TXER%)
2100    IF TXER%<>0 THEN IMPER%=2:LINEREP=2090:GOTO 6290
2110    DAV%=0
2120    WHILE IMPER%=0
2130       CLS:LOCATE 25,1
2140       PRINT TAB(53);"F1+F10 to EXIT"
2150       WHILE DAV%=0 AND IMPER%=0
2160          FOR ADR%=1 TO 2
2170             STR%=0
2180             RXER%=0
2190             CALL IMPTEST(ADR%,STR%,RXER%)
2200             IF RXER%=-1 THEN IMPER%=3:LINEREP=2190:GOTO 6290
2210             IF RXER%=1  THEN DAV%=1
2220          NEXT
2230       WEND
2240       IF DAV%=0 THEN 960
2250       '
2260       ' RESET BLOCK
2270       '
2280       FOR POD%=1 TO 2
2290          IF POD%=1 THEN CHA%=17 ELSE CHA%=2
2300          FOR I%=1 TO CHA%
2310             IMP.DATA(I%,POD%)=0
2320             STATUS%(I%,POD%)=0
2330             COUNT%(I%,1)=0:COUNT%(I%,2)=1
2340          NEXT
2350       NEXT
2360       N%=0
2370       A.SIN=0
2380       A.COS=0
2390       SX=0
2400       SY=0
2410       SCAN%=0
2420       MT%=0
2430       MK%=0
2440       '
2450       ' PRINT HEADING
2460       '
```

```
2470        COLOR 0,7
2480        LOCATE 3,20:PRINT "                                              "
2490        LOCATE 4,20:PRINT "  Data from IMP 35951C and IMP 35952A.  "
2500        LOCATE 5,20:PRINT "                                              "
2510        DMY$=FN DAY$(DATE$)
2520        LOCATE 9,28:PRINT "                          "
2530        LOCATE 10,28:PRINT "   Date : ";DMY$;"   "
2540        LOCATE 11,28:PRINT "                          "
2550        '
2560        ' START TIME SETTING
2570        '
2575 'GOTO 2873
2580        START$="HH:MM:SS"
2585        IF TIME$="24:00:00" OR FN TIM(TIME$)<18 THEN START$="00:00:00":GOTO
2587        TIM$=TIME$
2590        ON VAL(MID$(TIM$,5,1))+1 GOTO 2600,2600,2600,2600,2600,2630,2630,
2600        MID$(START$,5,1)="5"
2610        MID$(START$,1,4)=MID$(TIM$,1,4)
2620        GOTO 2870
2630        MID$(START$,5,1)="0"
2640        ON VAL(MID$(TIM$,4,1))+1 GOTO 2650,2650,2650,2650,2650,2710
2650        M$=MID$(TIM$,4,1)
2660        V=ASC(M$)
2670        V=V+1
2680        MID$(START$,4,1)=CHR$(V)
2690        MID$(START$,1,2)=MID$(TIM$,1,2)
2700        GOTO 2870
2710        MID$(START$,4,1)="0"
2720        ON VAL(MID$(TIM$,2,1))+1 GOTO 2730,2730,2730,2790,2730,2730,2730,
2730        H$=MID$(TIM$,2,1)
2740        V=ASC(H$)
2750        V=V+1
2760        MID$(START$,2,1)=CHR$(V)
2770        MID$(START$,1,1)=MID$(TIM$,1,1)
2780        GOTO 2870
2790        ON VAL(LEFT$(TIM$,1))+1 GOTO 2730,2730,2800
2800        MID$(START$,2,1)="0"
2810        ON VAL(LEFT$(TIM$,1))+1 GOTO 2820,2840,2860
2820        MID$(START$,1,1)="1"
2830        GOTO 2870
2840        MID$(START$,1,1)="2"
2850        GOTO 2870
2860        MID$(START$,1,1)="0"
2870        MID$(START$,7,2)="00"
2873        IF FIN%=1 THEN 3393
2880        LOCATE 17,10:PRINT"                    " :LOCATE 17,53:PRINT"
2890        LOCATE 18,10:PRINT"   Actual  time     " :LOCATE 18,53:PRINT"    Next
2900        LOCATE 19,10:PRINT"                    " :LOCATE 19,53:PRINT"
2910        LOCATE 20,10:PRINT"     ";TIME$;"      ":LOCATE 20,53:PRINT"     ";S
2920        LOCATE 21,10:PRINT"                    " :LOCATE 21,53:PRINT"
2923 'IF VAL(RIGHT$(TIME$,2)) <> 0 THEN 2880  'Alleen bij starten op hele minu
2924 'START$=TIME$
2930        IF KPR%=2 THEN 6630
2935        IF FN TIM(TIME$)<18 THEN 2960
2940        IF RIGHT$(TIME$,2)<>"00" THEN 2880
2950        IF VAL(MID$(TIME$,5,1))<>0 AND VAL(MID$(TIME$,5,1))<>5 THEN 2880
2960        COLOR 7,0
2970        B!=FN TIM(TIME$)
2990        IF B!>0 AND B!<86300! THEN 3030 ELSE CLOSE #1:CLOSE #3:GOTO 1510
3030        A!=FN TIM(TIME$)
3035        IMPTX$="CL1;CL2"
3040        TXER%=0
3050        CALL IMPTX(IMPAD(2),IMPTX$,TXER%)
3060        IF TXER%<>0 THEN IMPER%=2:LINEREP=3050:GOTO 6290
```

```
3070        END.TIME!=(FN TIM(START$))+INTERVAL
3075        FOR TEL=1 TO 1500:NEXT
3080        CLS
3090        '
3100        ' 5 MINUTES SCAN LOOP
3110        '
3120        WHILE A!<END.TIME!
3130           IF KPR%=2 THEN 6640
3140           A!=FN TIM(TIME$)
3150           SCAN%=SCAN%+1
3160           FOR POD%=1 TO 2
3170              ADR%=POD%
3180              STR%=0
3190              RXER%=0
3200              IF POD%=1 THEN CHA%=17 ELSE CHA%=2
3210              CALL IMPNUMERIC (ADR%,STR%,RES(1),STA%(1),CHA%,RXER%)
3220              IF RXER%<>0 THEN IMPER%=2:LINEREP=3210:GOTO 6290
3230              IF POD%=1 THEN 3260
3240              DELTA!=B!-A!
3250              B!=A!
3260              FOR I%=1 TO CHA%
3270                 IF STA%(I%)=-31 THEN 3320
3280                 IF POD%=2 THEN 3310
3290                 ON I% GOSUB 3600,3600,3882,3883,3660,3660,3884,3885,3730,3
3291                 ' channel    1    2    3    4    5    6    7    8    9
3300                 GOTO 3320
3310                 ON I% GOSUB 4000,4060
3311                 ' channel    1    2
3320              NEXT
3330           NEXT
3340           ON ACT%+1 GOSUB 5640,5330
3350        WEND
3360        '
3370        ' DATA ACQUISITION
3380        '
3390        FIN%=1:GOTO 2580
3393        IF START$="00:00:00" THEN FIN$="24:00:00" ELSE FIN$=START$
3394        FIN%=0
3397        POD%=1
3400        CHA%=17
3410        FOR I%=1 TO CHA%
3420           IF COUNT%(I%,POD%)=0 THEN COUNT%(I%,POD%)=1
3430           IMP.DATA(I%,POD%)=IMP.DATA(I%,POD%)/COUNT%(I%,POD%)
3440           STATUS%(I%,POD%)=INT(STATUS%(I%,POD%)/COUNT%(I%,POD%)*10)/10
3450        NEXT
3460        GOSUB 4430
3470        PRINT #1,DATE$;",";FIN$;",";SCAN%
3480        FOR M%=1 TO 8
3490           PRINT #1,USING"+####.##";METEO.ARRAY(M%);
3500           PRINT #1,",";
3510           PRINT #1,USING"###";NUMBER%(M%)
3520        NEXT
3521        PRINT #3,DATE$;",";FIN$;",";SCAN%
3522        FOR M%=9 TO 15
3523           PRINT #3,USING"+####.##";METEO.ARRAY(M%);
3524           PRINT #3,",";
3525           PRINT #3,USING"###";NUMBER%(M%)
3526        NEXT
3530     WEND
3540 WEND
3550 '
3560 '                                  SELECTION OF DATA , IMP 1C
3570 '
3580 '                                   T , MODE 500, 0-20 mA
```

```
3590 '                                      RH, MODE 500, 0-20 mA
3600 RAW(I%)    = RES(I%)
3610 SIGN%(I%) = STA%(I%)
3620 IF RES(I%) > 25   THEN 3950
3630                         GOTO 3860
3640 '                                      Airpr 1, MODE 100, 0-2.5 V
3650 '                                      Airpr 1, MODE 100, 0-2.5 V
3660 J%          = I%-2
3670 RAW(J%)    = RES(I%)
3680 SIGN%(J%) = STA%(I%)
3690 IF RES(I%) > 2.5  THEN 3950
3700                         GOTO 3860
3710 '                                      Wind.dir 1, MODE 100, 0-2.0 V
3720 '                                      Wind.dir 2, MODE 100, 0-2.0 V
3730 J%          = I%-4
3740 RAW(J%)    = RES(I%)
3750 SIGN%(J%) = STA%(I%)
3760 IF RES(I%) > 2!   THEN 3950
3770                         GOTO 3860
3780 '                                      PIR , MODE 100, 0-2 mV
3790 '                                      CM11, MODE 100, 0-2 mV
3800 RAW(7)    = RES(13)
3810 SIGN%(7) = STA%(13)
3820 RAW(8)    = RES(17)
3830 SIGN%(8) = STA%(17)
3840 IF RES(I%) > 1.5  THEN 3950
3850                         GOTO 3880
3860 IF RES(I%) < 0    THEN 3950
3870                         GOTO 3890
3880 IF RES(I%) < -1   THEN 3950
3881                         GOTO 3890
3882 RAW(11) = RES(3)  : SIGN%(11) = STA%(3)  : GOTO 3889
3883 RAW(12) = RES(4)  : SIGN%(12) = STA%(4)  : GOTO 3889
3884 RAW(13) = RES(7)  : SIGN%(13) = STA%(7)  : GOTO 3889
3885 RAW(14) = RES(8)  : SIGN%(14) = STA%(8)  : GOTO 3889
3886 RAW(15) = RES(11) : SIGN%(15) = STA%(11) : GOTO 3889
3887 RAW(16) = RES(15) : SIGN%(16) = STA%(15) : GOTO 3889
3888 RAW(17) = RES(16) : SIGN%(17) = STA%(16)
3889 IF RES(I%) < -100 OR RES(I%) > 100        THEN 3950
3890 IF RES(I%) = 0   AND STA%(I%) = 0         THEN 3950
3900 STATUS%(I%,1) = STATUS%(I%,1) + STA%(I%)
3910 COUNT%(I%,1)  = COUNT%(I%,1)  + 1
3920 IF I%=9  THEN GOSUB 4220
3930 IF I%=10 THEN GOSUB 4280
3940 IMP.DATA(I%,1)= IMP.DATA(I%,1)+ RES(I%)
3950 RETURN
3960 '
3970 '                                    SELECTION OF DATA , IMP 2A
3980 '
3990 '                                     Wind.Speed, MODE 740, 0-1E+5 c
4000 J%          = I% + 8
4010 RAW(J%)    = RES(I%)
4020 SIGN%(J%) = STA%(I%)
4030 IF RES(I%) > 1000000!  THEN  4150
4040                         GOTO 4110
4050 '                                     Precip, MODE 740, 0-1E+3 c
4060 J%          = I% + 8
4070 RAW(J%)    = RES(I%)
4080 SIGN%(J%) = STA%(I%)
4090 IF RES(I%) > 10000!    THEN  4150
4100 IF SCAN% < 4           THEN  4130
4110 IF RES(I%) <= 0        THEN  4150
4120 'IF I% = 1             THEN  GOSUB 4340  'this part is out of use
4130 COUNT%(I%,2)   = COUNT%(I%,2) + 1
```

```
4140 IMP.DATA(I%,2) = RES(I%)
4150 RETURN
4160 '
4170 '                               CONVERSION OF RAW DATA TO METEO DATA
4180 '                               ------------------------------------
4190 '
4200 '                                       WIND DIRECTION
4210 '
4220 ANGLE1 = 381.5 * RES(9) - 111.8    'was 389.3 * RES(9) - 287.71
4222 IF ANGLE1 > 360 THEN ANGLE1=ANGLE1-360
4230 A1.SIN = SIN (ANGLE1 / 57.2958)
4240 A1.COS = COS (ANGLE1 / 57.2958)
4250 A.SIN  = A.SIN + A1.SIN
4260 A.COS  = A.COS + A1.COS
4270                                 RETURN
4280 ANGLE2 = -380.9 * RES(10) + 841.6  'was -389.4 * RES(10) + 651.67
4282 IF ANGLE2 > 360 THEN ANGLE2=ANGLE2-360
4290 A2.SIN = SIN (ANGLE2 / 57.2958)
4300 A2.COS = COS (ANGLE2 / 57.2958)
4310 A.SIN  = A.SIN + A2.SIN
4320 A.COS  = A.COS + A2.COS
4330                                 RETURN
4340 FWS = IMP.DATA(1,2) / DELTA!        'this part is out of use
4350 WS  = .0494 * FWS + .47             'this part is out of use
4360 SX  = SX + WS * A1.COS + WS * A2.COS   'this part is out of use
4370 SY  = SY + WS * A1.SIN + WS * A2.SIN   'this part is out of use
4380                            RETURN   'this part is out of use
4390 '
4400 '                                       TEMPERATURE
4410 '
4420 '               TEMP.CAL (ºC/mA)
4430 POD%            = 1
4440 TEMP.CAL        = 3.5
4450 CH%             = 1
4460 NUMBER%(1)      = COUNT%(CH%,POD%)
4470 TEMP.CUR        = IMP.DATA(CH%,POD%)
4480 TEMP.DEG        = TEMP.CUR * TEMP.CAL - 25!
4490 METEO.ARRAY(1)= (INT(TEMP.DEG * 100)) / 100
4500 '
4510 '                                       RELATIVE HUMIDITY
4520 '
4530 '               RHUM.CAL (%/mA)
4540 RHUM.CAL        = 4
4550 CH%             = 2
4560 NUMBER%(2)      = COUNT%(CH%,POD%)
4570 RHUM.CUR        = IMP.DATA(CH%,POD%)
4580 RHUM.PRO        = RHUM.CUR * RHUM.CAL
4590 IF RHUM.PRO > 100 THEN RHUM.PRO = 100
4600 METEO.ARRAY(2)= (INT(RHUM.PRO * 100)) / 100
4610 '
4620 '                                       AIRPRESSURE
4630 '
4650 CH%             = 5
4660 NUMBER%(3)      = COUNT%(CH%,POD%)
4670 AIRP1.VOL       = IMP.DATA(CH%,POD%)
4680 CH%             = 6
4690 NUMBER%(3)      = (NUMBER%(3) + COUNT%(CH%,POD%)) / 2
4700 AIRP2.VOL       = IMP.DATA(CH%,POD%)
4710 PRES.5          = -87.635    * AIRP1.VOL + 1140.39
4720 PRES.6          =  91.42401 * AIRP2.VOL + 853.87
4730 AIRP.BAR        = (PRES.5 + PRES.6) / 2
4740 METEO.ARRAY(3)= (INT(AIRP.BAR * 100)) / 100
4750 '
4760 '                                       WIND DIRECTION
```

```
4770 '
4780 '               A.SIN + A.COS
4790 NUMBER%(4)     = COUNT%(9,1) + COUNT%(10,1)
4800 WIND.DIR       = ATN (A.SIN / A.COS) * 57.2958
4810 IF A.COS < 0 THEN 4840    'was 4830
4820 'IF A.SIN > 0 THEN 4840
4830 WIND.DIR = WIND.DIR + 180
4840 WIND.DIR = WIND.DIR + 180
4850 IF WIND.DIR > 360 THEN WIND.DIR = WIND.DIR - 360
4860 METEO.ARRAY(4) = (INT(WIND.DIR * 100)) / 100
4870 '
4880 '                                              LONG WAVE RADIATION
4890 '
4900 '               PIR.CAL (4.56 µV/Wm-2)
4910 PIR.CAL        = 4.56E-06
4920 CH%            = 13
4930 NUMBER%(5)     = COUNT%(CH%,POD%)
4940 PIR.VOL        = IMP.DATA(CH%,POD%)
4950 PIR.RAD        = PIR.VOL / PIR.CAL
4960 METEO.ARRAY(5)= (INT(PIR.RAD * 100)) / 100
4970 '
4980 '                                              SHORT WAVE RADIATION
4990 '
5000 '               CM11.CAL (4.86 µV/Wm-2)
5010 CM11.CAL       = 4.86E-06
5020 CH%            = 17
5030 NUMBER%(6)     = COUNT%(CH%,POD%)
5040 CM11.VOL       = IMP.DATA(CH%,POD%)
5050 CM11.RAD       = CM11.VOL / CM11.CAL
5060 METEO.ARRAY(6)= (INT(CM11.RAD * 100)) / 100
5061 '
5062 ' TEMPERATURE Cu-Co
5063 '
5064 'DIRECT TEMPERATURES, NO CONVERSION CH's 3,4,7,8,11,15,16
5065 CH%             = 3                 '1.5 m beside HMP123Y
5066 NUMBER%(9)      = COUNT%(CH%,POD%)
5067 METEO.ARRAY(9)  = (INT(IMP.DATA(CH%,POD%) * 100)) / 100
5068 CH%             = 4                 '-0.1 m in Al-pipe/ground
5069 NUMBER%(10)     = COUNT%(CH%,POD%)
5070 METEO.ARRAY(10)= (INT(IMP.DATA(CH%,POD%) * 100)) / 100
5071 CH%             = 7                 '-0.5 m in Al-pipe/ground
5072 NUMBER%(11)     = COUNT%(CH%,POD%)
5073 METEO.ARRAY(11)= (INT(IMP.DATA(CH%,POD%) * 100)) / 100
5074 CH%             = 8                 'surface concrete dispersal
5075 NUMBER%(12)     = COUNT%(CH%,POD%)
5076 METEO.ARRAY(12)= (INT(IMP.DATA(CH%,POD%) * 100)) / 100
5077 CH%             = 11                '+5 m in wind mast
5078 NUMBER%(13)     = COUNT%(CH%,POD%)
5079 METEO.ARRAY(13)= (INT(IMP.DATA(CH%,POD%) * 100)) / 100
5080 CH%             = 15                '+7.5 m in wind mast
5081 NUMBER%(14)     = COUNT%(CH%,POD%)
5082 METEO.ARRAY(14)= (INT(IMP.DATA(CH%,POD%) * 100)) / 100
5083 CH%             = 16                'testkapje pyrgeometer
5084 NUMBER%(15)     = COUNT%(CH%,POD%)
5085 METEO.ARRAY(15)= (INT(IMP.DATA(CH%,POD%) * 100)) / 100
5089 '
5090 ' WIND SPEED
5095 '
5100 'WISP.FRE - FREQ (Hz) - WISP.MPS (m/s)
5110 POD%           = 2
5120 CH%            = 1
5130 NUMBER%(7)     = COUNT%(CH%,POD%)
5140 WISP.FRE       = IMP.DATA(CH%,POD%)
5150 FREQ           = WISP.FRE / (INTERVAL - 3 * INTERVAL / SCAN%)
```

```
5160 WISP.MPS       = .04959 * FREQ + .279996
5170 METEO.ARRAY(7)= (INT(WISP.MPS * 100)) / 100
5180 '
5190 ' PRECIPITATION
5200 '
5210 'PULS - RAIN.FRE (min-1) - RAIN.MIN (mm/min)
5220 CH%            = 2
5230 NUMBER%(8)     = COUNT%(CH%,POD%)
5240 PULS           = IMP.DATA(CH%,POD%)
5250 RAIN.FRE       = 60 * PULS / (INTERVAL - 3 * INTERVAL / SCAN%)
5260 RAIN.MIN       = .0995 * RAIN.FRE + .0006# * RAIN.FRE^2
5270 PRECIP.MM      = (INTERVAL - 3 * INTERVAL / SCAN%) * RAIN.MIN / 60
5280 METEO.ARRAY(8)= (INT(PRECIP.MM * 100)) / 100
5290 RETURN
5300 '
5310 ' RAW DATA PER SCAN
5320 '
5330 IF KPR%=2 THEN 6640
5335 IF SCR%=1 THEN 5360
5340 CLS
5350 SCR%=1
5360 LOCATE 1,1
5370 COLOR 0,7
5380 PRINT " Meteo data of  ";DMY$
5390 COLOR 7,0
5400 'PRINT
5410 PRINT TAB(5);"IMP";TAB(10);"TYPE";TAB(31);"RESULT";TAB(51);"STATUS"
5420 PRINT "-------------------------------------------------------------------
5430 FOR J%=1 TO 8
5440    PRINT TAB(5);" 1 ";TAB(10);CHANNEL$(J%);TAB(30);RAW(J%);TAB(50);SIGN%(
5450 NEXT
5452 FOR J%=11 TO 17
5453    PRINT TAB(5);" 1 ";TAB(10);CHANNEL$(J%);TAB(30);RAW(J%);TAB(50);SIGN%(
5454 NEXT J%
5460 'COLOR 0,7
5470 'PRINT
5480 'PRINT " Meteo data from IMP 2 ";TAB(70);"     "
5490 'PRINT
5500 'COLOR 7,0
5510 FOR J%=9 TO 10
5520    PRINT TAB(5);" 2 ";TAB(10);CHANNEL$(J%);TAB(30);RAW(J%);TAB(51);SIGN%(
5530 NEXT
5540 'PRINT
5550 COLOR 0,7
5560 PRINT TAB(10);"TIME :";TAB(30);TIME$;TAB(51);"SCAN: ";SCAN%;TAB(70);"
5570 COLOR 7,0
5580 'PRINT:PRINT:PRINT
5590 PRINT TAB(10);"F2 to display METEO data";TAB(51);"F1+F10 to EXIT"
5600 RETURN
5610 '
5620 ' METEO DATA PER 5 MINUTES
5630 '
5640 IF KPR%=2 THEN 6640
5645 LOCATE 1,58
5650 PRINT"   TIME : ";TIME$
5660 IF MT%=1 THEN 5900
5670 CLS:COLOR 0,7
5680 PRINT" METEO DATA  of ";DMY$;" at ";FIN$;"                    "
5690 COLOR 7,0
5700 PRINT:PRINT
5710 PRINT USING"Temperature (ºC)         : +###.##";METEO.ARRAY(1)
5720 PRINT USING"           Cu-Co (- 0.5 m) :            +###.##";METEO.ARRAY(
5730 PRINT USING"Relative Humidity (%)     :  ###.##";METEO.ARRAY(2)
5740 PRINT USING"           Cu-Co (- 0.1 m) :            +###.##";METEO.ARRAY(
```

```
5750 PRINT USING"Airpressure (mbar)       : ####.##";METEO.ARRAY(3)
5760 PRINT USING"          Cu-Co (concrete):          +###.##";METEO.ARRAY(
5770 PRINT USING"0.3-3 µm Radiation (W/m2):+####.##";METEO.ARRAY(6)
5780 PRINT USING"          Cu-Co (+ 1.5 m) :          +###.##";METEO.ARRAY(
5790 PRINT USING"4-50 µm Radiation (W/m2) :+####.##";METEO.ARRAY(5)
5800 PRINT USING"          Cu-Co (+ 5 m)   :          +###.##";METEO.ARRAY(
5810 PRINT USING"Wind Speed (m/s)         :  ###.##";METEO.ARRAY(7)
5820 PRINT USING"          Cu-Co (+ 7.5 m) :          +###.##";METEO.ARRAY(
5830 PRINT USING"Wind Direction (º)       :  ###.##";METEO.ARRAY(4)
5840 PRINT USING"          Cu-Co (PIR-cup) :          +###.##";METEO.ARRAY(
5850 PRINT USING"Precipitation (mm/5 min) :  ###.##";METEO.ARRAY(8)
5860 LOCATE 24,1
5870 PRINT TAB(10);"F3 to display RAW data";TAB(53);"F1+F10 to EXIT";
5880 MK%=1
5890 MT%=1
5900 RETURN
5910 '
5920 ' KEY FUNCTION BLOCK 2
5930 '
5940 IF KPR%<>0  THEN 5960
5950 KPR%=2
5960 RETURN
5970 '
5980 ' ACTIVATE F2 KEY TO DISPLAY 5-MIN AVERAGES METEO DATA
5990 '
6000 ACT%=0
6010 SCR%=0
6020 RETURN
6030 '
6040 ' ACTIVATE F3 KEY TO DISPLAY RAW DATA
6050 '
6060 ACT%=1
6070 MK%=0
6080 MT%=0
6090 SCR%=0
6100 RETURN
6110 '
6120 ' ERROR TRAPPING AND RECOVERING
6130 '
6140 IF ERR=6  THEN ERREP=ERR:LINEREP=ERL:GOTO 6240
6150 IF ERR=11 THEN ERREP=ERR:LINEREP=ERL:GOTO 6240
6160 IF ERR=24 THEN ERREP=ERR:LINEREP=ERL:GOTO 6240
6170 IF ERR=25 THEN ERREP=ERR:LINEREP=ERL:GOTO 6240
6180 IF ERR=51 THEN ERREP=ERR:LINEREP=ERL:GOTO 6240
6190 IF ERR=57 THEN ERREP=ERR:LINEREP=ERL:GOTO 6240
6200 IF ERR=61 THEN ERREP=ERR:LINEREP=ERL:GOTO 6240
6210 IF ERR=69 THEN ERREP=ERR:LINEREP=ERL:GOTO 6240
6220 IF ERR=72 THEN ERREP=ERR:LINEREP=ERL:GOTO 6240
6230 GOTO 6440
6240 OPEN DRIVE$+"ERROR.MET" FOR APPEND AS #2
6250 PRINT #2,DMY$;",";TIME$;",SYSTEM,";ERREP;",";LINEREP
6260 CLOSE #2
6270 RESUME
6280 '
6290 OPEN DRIVE$+"ERROR.MET" FOR APPEND AS #2
6291 IF FOUND%<>0 THEN ERREP=150+FOUND%:GOTO 6300
6292 IF TXER%<>0  THEN ERREP=250+TXER% :GOTO 6300
6293 IF RXER%<>0  THEN ERREP=350+RXER% :GOTO 6300
6294 IF IMPER%<>0 THEN ERREP=450+IMPER%:GOTO 6300
6295 ERREP=999:LINEREP=99999!
6300 PRINT #2,DMY$;",";TIME$;",IMP,";ERREP;",";LINEREP
6310 CLOSE #2
6320 '
6330 ' RESET ERROR FLAG RXER% :-1 to 0 (IMPER%)
```

```
6340 '
6350 IF ADR%=2 THEN 6400
6360 N%=12
6370 A$=SPACE$(N%)
6380 E%=0
6390 CALL IMPSTRING(ADR%,STR%,A$,N%,E%)
6400 GOTO 1560
6410 '
6420 ' IRRECOVERABLE ERROR IN BASIC - STOP PROGRAMME
6430 '
6440 CLS:LOCATE 4,19
6450 PRINT "**********************************************"
6460 LOCATE 5,19
6470 PRINT "   A system error is found in line : ";ERL
6480 LOCATE 6,19
6490 PRINT "   The error number is             : ";ERR
6500 LOCATE 7,19
6510 PRINT "**********************************************"
6520 FOR I=1 TO 3000:NEXT
6530 IF ERR=61 OR ERR=67 THEN 6550
6540 RESUME 1250
6550 CLS
6560 LOCATE 4,15:PRINT "      - - - DISK FULL - - -          "
6570 LOCATE 6,15:PRINT "  PUT AN EMPTY DISK IN DRIVE D:     "
6577 LOCATE 7,15:PRINT "   RESTART METEO.BAS WITH F2        "
6580 KEY ON
6590 END
6600 '
6610 ' END OF METEO PROGRAMME
6620 '
6630 COLOR 7,0
6640 CLS
6650 COLOR 0,7
6655 ENDFILE$=MID$(DMY$,4,3)+LEFT$(DMY$,2)+RIGHT$(DMY$,2)
6660 LOCATE 6,23  : PRINT "                                   "
6670 LOCATE 7,23  : PRINT "  Last METEO-DATA stored in file   "
6680 LOCATE 8,23  : PRINT "                                   "
6690 LOCATE 9,23  : PRINT "            ";ENDFILE$+"M";"                "
6700 LOCATE 10,23 : PRINT "                                   "
6701 LOCATE 11,23 : PRINT "  Last THERMOCOUPLE-DATA in file   "
6702 LOCATE 12,23 : PRINT "                                   "
6703 LOCATE 13,23 : PRINT "            ";ENDFILE$+"T";"                "
6704 LOCATE 14,23 : PRINT "                                   "
6710 LOCATE 15,23 : PRINT "        End of programme.          "
6720 LOCATE 16,23 : PRINT "                                   "
6730 COLOR 7,0
6740 LOCATE 24,1
6750 KEY ON
6760 END
```

UNCLASSIFIED

REPORT DOCUMENTATION PAGE (MOD-NL)

1. DEFENSE REPORT NUMBER (MOD-NL) TD91-3894	2. RECIPIENT'S ACCESSION NUMBER	3. PERFORMING ORGANIZATION REPORT NUMBER FEL-91-B202
4. PROJECT/TASK/WORK UNIT NO. 22173	5. CONTRACT NUMBER -	6. REPORT DATE NOVEMBER 1991
7. NUMBER OF PAGES 35 (INCLUDING APPENDICES & RDP, EXCLUDING DISTRIBUTION LIST)	8. NUMBER OF REFERENCES -	9. TYPE OF REPORT AND DATES COVERED FINAL REPORT

10. TITLE AND SUBTITLE
AN AUTOMATIC METEORLOGICAL STATION

11. AUTHOR(S)
ING. R. VAN DER TOUW

12. PERFORMING ORGANIZATION NAME(S) AND ADDRESS(ES)
TNO PHYSICS AND ELECTRONICS LABORATORY, P.O. BOX 96864, 2509 JG THE HAGUE
OUDE WAALSDORPERWEG 63, THE HAGUE, THE NETHERLANDS

13. SPONSORING/MONITORING AGENCY NAME(S)
TNO PHYSICS AND ELECTRONICS LABORATORY,
THE HAGUE, THE NETHERLANDS

14. SUPPLEMENTARY NOTES

15. ABSTRACT (MAXIMUM 200 WORDS, 1044 POSITIONS)
FEL-TNO OWNS AN AUTOMATIC WEATHER STATION TO SUPPORT IR MEASUREMENTS IN THE FIELD. AFTER INSTALLATION THIS STATION AUTONOMOUSLY COLLECTS AVERAGED DATA. THE AVERAGE IS TAKEN OVER THE LAST 5 MINUTES INTERVAL AND STORED. THE STATION CAN RUN UNATTENDED FOR AT LEAST A WEEK.

16. DESCRIPTORS MEASURING INSTRUMENTS METEOROLOGICAL INSTRUMENTS	IDENTIFIERS

17a. SECURITY CLASSIFICATION (OF REPORT) UNCLASSIFIED	17b. SECURITY CLASSIFICATION (OF PAGE) UNCLASSIFIED	17c. SECURITY CLASSIFICATION (OF ABSTRACT) UNCLASSIFIED
18. DISTRIBUTION/AVAILABILITY STATEMENT UNLIMITED		17d. SECURITY CLASSIFICATION (OF TITLES) UNCLASSIFIED

UNCLASSIFIED

METEO Output example

```
04-01-1990,00:05:00, 49        date
  -24.5,  5                    time
   +0.0,  2                    number of scans in last period
+1029.4, 49                    meteo data block (8 parameters)
  +88.0, 98                       meteo parameter i
 +283.7, 49                       number of valid scans of parameter i
   +0.0, 49                    end block (8)
   +2.9, 48
   +0.0,  4
04-01-1990,00:10:00, 52
  -24.5,  1
   +0.0,  1
+1029.4, 52
  +88.7,104
 +283.6, 52
   +0.0, 52
   +3.3, 52
   +0.0,  4
04-01-1990,00:15:00, 52
  -24.5,  3
   +0.0,  4
+1029.4, 52
  +84.6,104
 +283.9, 52
   +0.0, 52
   +2.9, 52
   +0.0,  4
04-01-1990,00:20:00, 52
  -24.5,  4
   +0.0,  1
+1029.4, 52
  +87.4,104
 +283.5, 52
   +0.0, 52
   +2.7, 52
   +0.0,  4
04-01-1990,00:25:00, 52
  -24.5,  1
   +0.0,  1
+1029.4, 52
  +89.6,104
 +283.0, 52
   +0.0, 52
   +2.9, 52
   +0.0,  4
04-01-1990,00:30:00, 52
  -24.5,  6
   +0.0,  1
+1029.5, 52
  +89.1,104
 +282.6, 52
   +0.0, 52
   +3.1, 52
   +0.0,  4
```